Finding the MEK$_L$ Performance

Definitions

Definitions

```
ID = {{1, 0}, {0, 1}};
X = {{0, 1}, {1, 0}};
Z = {{1, 0}, {0, -1}};
Y = ⅈ * X.Z;
HL = Cos[θ] * X + Sin[θ] Z;
HLη[xη_] = HL.MatrixPower[ⅈ * Y, xη];
```

Defining some matrices

```
ρONE[ ϵ_] = (ID + (1 - 2 ϵ) * HL) / 2;
ρPerfectGlobal = KroneckerProduct[(ID + HL) / 2, (ID + HL) / 2];
ρPerfectLocal1 = KroneckerProduct[ID, (ID + HL) / 2];
ρPerfectLocal2 = KroneckerProduct[ID, (ID + HL) / 2];
ρ[ ϵ_] = KroneckerProduct[ρONE[ ϵ], ρONE[ ϵ]];
```

We think of MEK being a protocol that takes 2 magic states in the state $\rho[\epsilon]$ and pumps on them with a circuit using a further 8 copies.

Channels

Channels

```
U1[x1_, x2_, x3_, x4_, x5_, x6_, x7_, x8_] =
  MatrixPower[KroneckerProduct[X, X], x1 + x3].
   MatrixPower[KroneckerProduct[Z, Z], x2 + x4];
U2[x1_, x2_, x3_, x4_, x5_, x6_, x7_, x8_] =
  MatrixPower[KroneckerProduct[X, X], x5 + x7].
   MatrixPower[KroneckerProduct[Z, Z], x6 + x8];
Proj[x1_, x2_, x3_, x4_, x5_, x6_, x7_, x8_, xη_] =
  KroneckerProduct[ID, HLη[xη]].(KroneckerProduct[ID, ID] +
      (-1)^(x3+x4+x5+x6) KroneckerProduct[HLη[xη], HLη[xη]]) / 2;

Kraus[x1_, x2_, x3_, x4_, x5_, x6_, x7_, x8_, xη_] :=
  U2[x1, x2, x3, x4, x5, x6, x7, x8].
   Proj[x1, x2, x3, x4, x5, x6, x7, x8, xη].U1[x1, x2, x3, x4, x5, x6, x7, x8];
```

We will apply a noisy channel to $\rho[\epsilon]$ which is a sum over Kraus operators labelled by [x1_,x2_,x3_,x4_,x5_,x6_,x7_,x8_], where
xj=1 indicates an error on the j^th of te 8 pumping magic states.

For each Kraus operator we further break it up into 3 parts U1, Proj and U2.
U1 and U2 are unitaries that consist of Pauli errors.
Proj is a noisy parity projection, which is usually a projection onto the even parity Hadamard space, though some Pauli errors can generate a projection onto the odd parity Hadamard space.

```
Channel[x1_, x2_, x3_, x4_, x5_, x6_, x7_, x8_, xη_, ρin_] :=
 Kraus[x1, x2, x3, x4, x5, x6, x7, x8, xη].ρin.
  ConjugateTranspose[Kraus[x1, x2, x3, x4, x5, x6, x7, x8, xη]]
```

The above code simply takes a Kraus operator and makes a quantum channel from it

```
w0 = Permutations[{0, 0, 0, 0, 0, 0, 0, 0}];
w2 = Permutations[{1, 1, 0, 0, 0, 0, 0, 0}];
w4 = Permutations[{1, 1, 1, 1, 0, 0, 0, 0}];
w6 = Permutations[{1, 1, 1, 1, 1, 1, 0, 0}];
w8 = {{1, 1, 1, 1, 1, 1, 1, 1}};
w = Union[w0, w2, w4, w6, w8];
size = Dimensions[w][[1]];
```

We construct a list of all nontrivial even weight errors (odd weight errors are always detected by the circuit)

```
FullChannel[ϵ_, δ_, ρin_] :=
 ((1 - δ) * Sum[(ϵ/(1 - ϵ))^Total[w[[k]]] * (1 - ϵ)^8 * Channel[w[[k, 1]], w[[k, 2]], w[[k, 3]],
      w[[k, 4]], w[[k, 5]], w[[k, 6]], w[[k, 7]], w[[k, 8]], 0, ρin],
     {k, 1, size}]) + (δ * Sum[(ϵ/(1 - ϵ))^Total[w[[k]]] * (1 - ϵ)^8 *
     Channel[w[[k, 1]], w[[k, 2]], w[[k, 3]], w[[k, 4]], w[[k, 5]],
      w[[k, 6]], w[[k, 7]], w[[k, 8]], 1, ρin], {k, 1, size}]);
```

Above we form a channel by summing over all errors of even weight (those not detected) and weight them by $\epsilon^N * (1-\epsilon)^{N-8}$ where N is the number of errors (the number of nonzero values for [x1_,x2_,x3_,x4_,x5_,x6_,x7_,x8_],)

Output state

```
ρOUT[ϵ3_, ϵk_, δ_] = FullChannel[ϵ3, δ, ρ[ϵk]];
```

We find ρOUT[ϵ] by applying the noisy map FullChannel[ϵ, ρin] to the input ρ[ϵ]. Remember the input consists of 2 noisy magic states and 8 noisy magic were responsible for implementing the noisy map.

Polynomials

```
Polynomials
```

```
Psuc[ ϵ3_, ϵk_, δ_] =
 Simplify[Simplify[Simplify[Tr[ρOUT[ ϵ3, ϵk, δ] /. {Conjugate[Sin[θ]] → Sin[θ],
         Conjugate[Cos[θ]] → Cos[θ]}] /.
      {Conjugate[Sin[θ]] → Sin[θ], Conjugate[Cos[θ]] → Cos[θ]}]] /.
   {Conjugate[Sin[θ]] → Sin[θ], Conjugate[Cos[θ]] → Cos[θ]}]
```

$$1 - 448\,\epsilon 3^5 + 448\,\epsilon 3^6 - 256\,\epsilon 3^7 + 64\,\epsilon 3^8 - \frac{1}{2}\,\delta\,(1 - 2\,\epsilon 3)^4\,(1 - 2\,\epsilon k)^2 - 2\,\epsilon k + 2\,\epsilon k^2 - 64\,\epsilon 3^3\left(2 - \epsilon k + \epsilon k^2\right) + 32\,\epsilon 3^4\left(9 - \epsilon k + \epsilon k^2\right) - 8\,\epsilon 3\left(1 - 2\,\epsilon k + 2\,\epsilon k^2\right) + 8\,\epsilon 3^2\left(5 - 6\,\epsilon k + 6\,\epsilon k^2\right)$$

The success probability.

```
ϵout[ ϵ3_, ϵk_, δ_] = 1/Psuc[ ϵ3, ϵk, δ] * Expand[Simplify[
    Simplify[Tr[ρOUT[ ϵ3, ϵk, δ].(KroneckerProduct[ID, ID] - ρPerfectLocal1)]  /.
       {Conjugate[Sin[θ]] → Sin[θ], Conjugate[Cos[θ]] → Cos[θ]}] /.
     {Conjugate[Sin[θ]] → Sin[θ], Conjugate[Cos[θ]] → Cos[θ]}]]
```

$$\left(\frac{\delta}{4} - 2\,\delta\,\epsilon 3 + 8\,\epsilon 3^2 + 6\,\delta\,\epsilon 3^2 - 48\,\epsilon 3^3 - 8\,\delta\,\epsilon 3^3 + 136\,\epsilon 3^4 + 4\,\delta\,\epsilon 3^4 - 224\,\epsilon 3^5 + 224\,\epsilon 3^6 - 128\,\epsilon 3^7 + 32\,\epsilon 3^8 + \epsilon k^2 - \delta\,\epsilon k^2 - 8\,\epsilon 3\,\epsilon k^2 + 8\,\delta\,\epsilon 3\,\epsilon k^2 + 24\,\epsilon 3^2\,\epsilon k^2 - 24\,\delta\,\epsilon 3^2\,\epsilon k^2 - 32\,\epsilon 3^3\,\epsilon k^2 + 32\,\delta\,\epsilon 3^3\,\epsilon k^2 + 16\,\epsilon 3^4\,\epsilon k^2 - 16\,\delta\,\epsilon 3^4\,\epsilon k^2\right) \Big/ \left(1 - 448\,\epsilon 3^5 + 448\,\epsilon 3^6 - 256\,\epsilon 3^7 + 64\,\epsilon 3^8 - \frac{1}{2}\,\delta\,(1 - 2\,\epsilon 3)^4\,(1 - 2\,\epsilon k)^2 - 2\,\epsilon k + 2\,\epsilon k^2 - 64\,\epsilon 3^3\left(2 - \epsilon k + \epsilon k^2\right) + 32\,\epsilon 3^4\left(9 - \epsilon k + \epsilon k^2\right) - 8\,\epsilon 3\left(1 - 2\,\epsilon k + 2\,\epsilon k^2\right) + 8\,\epsilon 3^2\left(5 - 6\,\epsilon k + 6\,\epsilon k^2\right)\right)$$

Error probability on a single output qubit.

```
Expand[Psuc[ ϵ, ϵ, 0]]
Series[ϵout[ ϵ, ϵ, 0], {ϵ, 0, 10}]
```

$$1 - 10\,\epsilon + 58\,\epsilon^2 - 192\,\epsilon^3 + 400\,\epsilon^4 - 544\,\epsilon^5 + 480\,\epsilon^6 - 256\,\epsilon^7 + 64\,\epsilon^8$$

$$9\,\epsilon^2 + 34\,\epsilon^3 - 22\,\epsilon^4 - 720\,\epsilon^5 - 2756\,\epsilon^6 + 1144\,\epsilon^7 + 56\,056\,\epsilon^8 + 227\,072\,\epsilon^9 - 31\,472\,\epsilon^{10} + O[\epsilon]^{11}$$

Above equations match the results of MEK

```
Normal[Series[Psuc[ ϵ3 * t, ϵk * t, δ * t], {t, 0, 1}]] /. {t → 1}
Normal[Series[ϵout[ ϵ3 * t, ϵk * t, δ * t^2], {t, 0, 2}]] /. {t → 1}
```

$$1 - \frac{\delta}{2} - 8\,\epsilon 3 - 2\,\epsilon k$$

$$\frac{\delta}{4} + 8\,\epsilon 3^2 + \epsilon k^2$$

The leading order $P_{\text{suc}}$and ϵ_{out}.